A L' Hospital -szabály

A L' Hospital szabály

Függvényvizsgálat során gyakran kerülünk szembe azzal a kérdéssel, hogy két olyan függvény hányadosának határértékét kell megállapítanunk, amelyeknek külön-külön 0 a határértéke a tekintett helyen., másképpen szólva $\frac{0}{0}$ illetve $\frac{\infty}{\infty}$ alakú kifejezések határértékét kell kiszámítanunk. Máskor külön- külön ∞-be tartó függvények hányadosának viselkedését kell leírnunk. De a következõ határozatlan alakokat is kezelnünk kell: $0\ \infty$, $\infty - \infty$, 1^{∞}, 0^0, ∞^0.
Ezekben az esetekben hasznos a következõ tétel nyújtotta szabály:

Tétel: Legyen az f és a g olyan egyváltozós valós függvény, amelyekre a következõk teljesülnek:

1) Az f és a g az a hely valamely környezetében (kivéve esetleg az a pontot) értelmezve van és ott differenciálható;
2) a g differenciálhányadosa ebben a környezetben nem nulla, azaz

$$\frac{d}{dx}\, g \neq 0;$$

3) $$\lim_{x\to a} f(x) = \lim_{x\to a} g(x) = 0 \text{ vagy}$$

$$\lim_{x\to a} |f(x)| = \lim_{x\to a} |g(x)| = \infty$$

4) Az $x = a$ helyen létezik a deriváltak hányadosának határértéke, azaz a

$$\lim_{x\to a} \left(\frac{\frac{d}{dx} f(x)}{\frac{d}{dx} g(x)} \right) = \text{L'} \quad \text{határérték.}$$

Ekkor létezik a

$$\lim_{x\to a} \left(\frac{f(x)}{g(x)} \right) = L \quad \text{határérték is és}$$

a kéthatárérték egyenlõ, L=L' ,azaz

$$\lim_{x\to a} \left(\frac{f(x)}{g(x)} \right) = \lim_{x\to a} \left(\frac{\frac{d}{dx} f(x)}{\frac{d}{dx} g(x)} \right)$$

Megjegyzés: A L' Hospital-szabály a végtelenben (minusz végtelenben) vett határérték kiszámítására is alkalmazható. A tételben ekkor az a helyére mindenütt
∞ $(-\infty)$ írandó, a feltételeknek pedig egy bizonyos K-nál nagyobb (kisebb) értékekre kell teljesülniük.

Példák, a Hosp eljárás

1.Példa: Számítsuk ki a L' Hospital szabály segítségével a következõ határértéket:

$$\lim_{x\to 0}\left(\frac{\sin(2\,x)}{e^{(-3\,x)}-1}\right)$$

Megoldás: Vizsgáljuk meg elõször, teljesítik-e a függvények a L' Hospital szabály alkalmazásának feltételeit!

```
> restart:
> f:=x->sin(2*x);
```

$$f := x \to \sin(2\,x) \qquad (2.1)$$

```
> g:=x->exp(-3*x)-1;
```

$$g := x \to e^{(-3\,x)} - 1 \qquad (2.2)$$

```
> plot([f(x),g(x)],x=-1..1,-2..2);
```




```
> readlib(iscont);
```

proc (*f*::*algebraic*, *z*::(*name* = *range*), *opt*) ...**end proc** (2.3)

1.) Megmutatjuk, hogy a határérték $\frac{0}{0}$ alakú:

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x\to 0} \sin(2x) = 0 \tag{2.4}$$

```
> Limit(g(x),x=0)=limit(g(x),x=0);
```

$$\lim_{x\to 0} \left(e^{(-3x)} - 1\right) = 0 \tag{2.5}$$

2) Megmutatjuk, hogy az f és a g függvény egyaránt differenciálható az $x_0 = 0$ egy környezetében.
Valójában többet tudunk mondani: Az iscont függvény-eljárás megvizsgálja, hogy a deriváltak folytonosak-e az adott intervallumon!

```
> `Folytonos-e az f deriváltja`=iscont(D(f)(x),x=-0.2..0.2);
```

$$\textit{Folytonos-e az f deriváltja} = \textit{true} \tag{2.6}$$

```
> `Folytonos-e a g deriváltja`=iscont(D(g)(x),x=-0.2..0.2);
```

$$\textit{Folytonos-e a g deriváltja} = \textit{true} \tag{2.7}$$

3) Most megnézzük, hogy teljesül-e a $\frac{d}{dx}\, g(x_0) \neq 0$ feltétel, Megnézzük, van -e zérushely egy, az $x_0 = 0$ helyet tartalmazó alkalmas intervallumban:(kivéve az x_0-t)

```
> x0:=0:
```

```
> dgx:={fsolve(D(g)(x),x=-0.2..0.2)};
```

$$dgx := \{\ \} \tag{2.8}$$

```
> zhely:=dgx minus {x0};
```

$$zhely := \{\ \} \tag{2.9}$$

```
> if nops(zhely)=0 then print(`A nevezõ deriváltja nem 0 az x0 egy környezetében`) fi;
```

$$\textit{A nevezõ deriváltja nem 0 az x0 egy környezetében} \tag{2.10}$$

Ezt erõsíti meg a $\frac{d}{dx}\, g(x)$ ábrája is:

```
> plot(D(g)(x),x=-0.2..0.2);
```



A feltételek mind teljesültek, tehát: Ha a deriváltak hányadosának létezik határértéke, akkkor ez megegyezik az $\frac{f(x)}{g(x)}$ hányados határértékével:

```
> Limit(D(f)(x)/D(g)(x),x=0)=limit(D(f)(x)/D(g)(x),x=0);
```

$$\lim_{x \to 0} \left(-\frac{2}{3} \frac{\cos(2x)}{e^{(-3x)}} \right) = \frac{-2}{3} \tag{2.11}$$

Tehát

```
>  Limit(f(x)/g(x),x=0)=limit(D(f)(x)/D(g)(x),x=0);
```

$$\lim_{x \to 0} \left(\frac{\sin(2x)}{e^{(-3x)} - 1} \right) = \frac{-2}{3} \tag{2.12}$$

Vizsgáljuk meg mindezt grafikusan is!

```
> x0:=0:
> `érintõ`:=proc(x0,f)
> global x;
> D(f)(x0)*(x-x0)+f(x0);
> end;
```

$érintõ := \mathbf{proc}\,(x0, f)\ \mathbf{global}\, x;\ \ (\mathrm{D}(f))(x0) * (x - x0) + f(x0)\ \mathbf{end\ proc}$ (2.13)

```
> `érintõ1 egyenlete`:=y=`érintõ`(x0,f);
```

$érintõ1\ egyenlete := y = 2\,x$ (2.14)

```
> erinto1:=unapply(evalf(rhs(%)),x);
```

$erinto1 := x \rightarrow 2.\,x$ (2.15)

```
> `érintõ2 egyenlete`:=y=`érintõ`(x0,g);
```

$érintõ2\ egyenlete := y = -3\,x$ (2.16)

```
> erinto2:=unapply(evalf(rhs(%)),x);
```

$erinto2 := x \rightarrow -3.\,x$ (2.17)

```
> 
> plot([f(x),erinto1(x),g(x),erinto2(x)],x=-0.51..2,y=-2..3,title=
  `A függvények és érintõik`,color=[red,blue],scaling=constrained,
  thickness=[2,2]);
```




```
> 
```

Láthatóan az érintõk meredekségének hányadosa éppen a kérdéses határértéket adja!

Sokszor a deriváltak hányadosa is $\frac{0}{0}$ alakú. Ekkor a szabály ismételt alkalmazásával juthatunk eredményre:

2. Példa: Számítsuk ki a L' Hospital szabály segítségével a következõ határértéket:

$$\lim_{x\to 0}\left(\frac{\tan(x)-x}{x-\sin(x)}\right)$$

```
> f:=x->tan(x)-x;
```

$$f := x \to \tan(x) - x \tag{2.18}$$

```
> g:=x->x-sin(x);
```

$$g := x \to x - \sin(x) \tag{2.19}$$

Esetünkben az elsõ és a második deriváltak is 0 értékûek az x=0 helyen és mivel folytonosak is, a határérték továbbra is $\frac{0}{0}$ alakú

```
> D(f)(0);D(g)(0);
```

$$0$$
$$0 \tag{2.20}$$

```
> (D@@2)(f)(0);(D@@2)(g)(0);
```

$$0$$
$$0 \tag{2.21}$$

A harmadik deriváltak már nem nulla értékûek az x = 0 helyen és mivel folytonosak , helyettesítési értékük hányadosa adja a határértéket:

```
> (D@@3)(f)(0);
> (D@@3)(g)(0);
```

$$2$$
$$1 \tag{2.22}$$

```
> Limit(f(x)/g(x),x=0)=limit((D@@3)(f)(x)/(D@@3)(g)(x),x=0);
```

$$\lim_{x\to 0}\left(\frac{\tan(x)-x}{x-\sin(x)}\right)=2 \tag{2.23}$$

```
>
```

A **Hosp** eljárás animációval szemlélteti, mi történik, akkor, amikor többször kell deriválni!

```
> Hosp:=proc(f,g,x0,h)
>  local i,k,pont:
>   i:=0:
>   if not(type(limit(f(x)/g(x),x=x0),numeric))  then    lprint
  (`Az`,x0,`pontban nincs határérték `):
>  else
>   while limit((D@@i)(f)(x),x=x0)=0 and limit((D@@i)(g)(x),x=x0)=
  0 do
>     i:=i+1
```

```
> od:
> for k from 0 to i-1 do
>  kep2||k:=plot([(D@@k)(g)(x),(D@@(k+1))(g)(x0)*(x-x0)+(D@@k)(g)
  (x0)],x=x0-h..x0+h,-h..h,color=[blue,red],thickness=[2,3]):
> kep1||k:=plot([(D@@k)(f)(x),(D@@(k+1))(f)(x0)*(x-x0)+(D@@k)(f)
  (x0)],x=x0-h..x0+h,-h..h,color=[black,red],thickness=[2,3]):
> pont:=plot([[x0,f(x0)]],style=point,symbol=circle,color=black):
> rajz||k:=plots[display]({kep1||k,kep2||k,pont}):
> od:
> plots[display]([rajz||(0..(i-1))],view=[x0-h..x0+h,-h..h],
  insequence=true);
> fi:
> end:
> x0:=0:
> f:=x->tan(x)-x;
> g:=x->x-sin(x);
```

$$f := x \to \tan(x) - x$$

$$g := x \to x - \sin(x) \qquad (2.24)$$

```
> Hosp(f,g,0,1);
```



Az animáció végrehajtásakor látjuk, hogy az elsõ és második deriváltak is 0 értékûek, ekkor a függvények ill. a deriváltak közös érintõje az $y = 0$ egyenes, az x tengely egyenese.Amásodik deriváltak érintõinek meredekségei, harmadik deriváltak már nem nulla értékûek. Az is látható, hogy az f és a g a 0 közelében "jól" megközelíti a nullát és a két görbe egymást is.Ez még szembetûnõbb, ha összehasonlítjuk viselkedésüket az 1. Példában szereplõ függvényekkel:

```
>
> f:=x->sin(2*x);
```

$$f := x \rightarrow \sin(2\,x) \tag{2.25}$$

```
> g:=x->exp(-3*x)-1;
```

$$g := x \rightarrow e^{(-3\,x)} - 1 \tag{2.26}$$

```
> Hosp(f,g,0,1);
```



>

Definíció: Akkor mondjuk, hogy az f és a g függvény az $x = x_0$ helyen **n-edrendben érinti egymást**, ha $f(x_0) = g(x_0)$ és a két függvény elsõ n differenciálhányadosa az x_0 helyen egyenlõ, de az $(n+1)$ -edik deriváltjuk már különbözõ.

Megállapíthatjuk, hogy az $f(x) = \sin(2x)$ és a $g(x) = \mathrm{e}^{(-3x)} - 1$ metszik (0-ad rendben érintik) egymást az origóban, míg az $f(x) = \tan(x) - x$ és a $g(x) = x - \sin(x)$ másodrendben érintik egymást az origóban.

3. Példa Vizsgáljuk meg a következõ határértéket:

$$\lim_{x\to 0} \left(\frac{\sqrt{x+1} - 1 - \frac{x}{2}}{x^2} \right)$$

Megoldás:

```
> f:=x->sqrt(1+x)-1-x/2;
```

$$f := x \to \sqrt{x+1} - 1 - \frac{1}{2}\, x \tag{2.27}$$

```
> g:=x->x^2;
```

(2.28)

$$g := x \rightarrow x^2 \qquad (2.28)$$

1.) Megmutatjuk, hogy a határérték $\frac{0}{0}$ alakú:

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \rightarrow 0} \left(\sqrt{x+1} - 1 - \frac{1}{2} x \right) = 0 \qquad (2.29)$$

```
> Limit(g(x),x=0)=limit(g(x),x=0);
```

$$\lim_{x \rightarrow 0} \left(x^2 \right) = 0 \qquad (2.30)$$

2) Megmutatjuk, hogy az f és a g függvény egyaránt differenciálható az $x_0 = 0$ egy környezetében. Valójában többet tudunk mondani: Az iscont függvény-eljárás megvizsgálja, hogy a deriváltak folytonosak-e az adott intervallumon!

```
> `Folytonos-e az f deriváltja`=iscont(D(f)(x),x=-0.2..0.2);
```

$$\textit{Folytonos-e az f deriváltja} = \textit{true} \qquad (2.31)$$

```
> `Folytonos-e a g deriváltja`=iscont(D(g)(x),x=-0.2..0.2);
```

$$\textit{Folytonos-e a g deriváltja} = \textit{true} \qquad (2.32)$$

3) Most megnézzük, hogy teljesül-e a $\frac{\partial}{\partial x} g\left(x_0\right) \neq 0$ feltétel, Megnézzük, van -e zérushely egy, az $x_0 = 0$ helyet tartalmazó alkalmas intervallumban:(kivéve az x_0-t:

```
> dgx:={fsolve(D(g)(x),x=-0.2..0.2)};
```

$$dgx := \{0.\} \qquad (2.33)$$

```
> x0:=0.;
```

$$x0 := 0. \qquad (2.34)$$

```
> zhely:=dgx minus {x0};
```

$$zhely := \{\ \} \qquad (2.35)$$

```
>
```

```
> if nops(zhely)=0 then print(`A nevezõ deriváltja nem 0 az x0 egy
  környezetében`)  fi;
```

$$\textit{A nevezõ deriváltja nem 0 az x0 egy környezetében} \qquad (2.36)$$

Ezt erõsíti meg a $\frac{d}{dx} g(x)$ ábrája is:

```
> plot(D(g)(x),x=-0.2..0.2);
```



A feltételek mind teljesültek, tehát: Ha a deriváltak hányadosának létezik határértéke, akkkor ez megegyezik az $\frac{f(x)}{g(x)}$ hányados határértékével.
A deriváltak is 0-hoz tartanak:

```
> Limit(D(f)(x),x=x0)=limit(D(f)(x),x=x0);
```

$$\lim_{x \to 0.} \left(\frac{1}{2} \frac{1}{\sqrt{x+1}} - \frac{1}{2} \right) = 0. \tag{2.37}$$

```
> Limit(D(g)(x),x=x0)=limit(D(g)(x),x=x0);
```

$$\lim_{x \to 0.} (2\,x) = 0. \tag{2.38}$$

Képezzük ezért a második deriváltakat s nézzük meg a határérékùket:

```
> Limit((D@@2)(f)(x),x=x0)=limit((D@@2)(f)(x),x=x0);
```

$$\lim_{x \to 0.} \left(-\frac{1}{4} \frac{1}{(x+1)^{(3/2)}} \right) = -.2500000000 \tag{2.39}$$

```
> Limit((D@@2)(g)(x),x=x0)=limit((D@@2)(g)(x),x=x0);
```

$$\lim_{x \to 0.} 2 = 2 \tag{2.40}$$

Ezek már nem 0 értékûek, tehát

```
> Limit(f(x)/g(x),x=0)=limit((D@@2)(f)(x)/(D@@2)(g)(x),x=0);
```

$$\lim_{x\to 0}\left(\frac{\sqrt{x+1}-1-\frac{1}{2}x}{x^2}\right)=\frac{-1}{8} \tag{2.41}$$

Tehát

```
> Limit(f(x)/g(x),x=0)=limit((D@@2)(f)(x)/(D@@2)(g)(x),x=0);
```

$$\lim_{x\to 0}\left(\frac{\sqrt{x+1}-1-\frac{1}{2}x}{x^2}\right)=\frac{-1}{8} \tag{2.42}$$

Szemléljük meg mindezt a Hosp eljárással:

```
> Hosp(f,g,0,1);
```



4. Példa: Számítsuk ki a

$$\lim_{x\to\infty}\left(\frac{e^x}{x^2}\right)$$

határértéket!

Megoldás:
A számláló és a nevezõ végtelenben vett határértéke egyaránt végtelen.Mindkét függvény minden valós x értékre differenciálható és a nevezõ deriváltja pozitív x értékekre nem egyenlõ 0-val. Ezért a szabály alkalmazható.

```
> f:=x->exp(x);
> g:=x->x^2;
```

$$f := x \to e^x$$

$$g := x \to x^2 \tag{2.43}$$

```
> Limit(f(x)/g(x),x=infinity)=Limit(D(f)(x)/D(g)(x),x=infinity);
```

$$\lim_{x \to \infty} \left(\frac{e^x}{x^2}\right) = \lim_{x \to \infty} \left(\frac{1}{2}\,\frac{e^x}{x}\right) \tag{2.44}$$

A kapott kifejezés is $\frac{\infty}{\infty}$ alakú, alkalmazzuk ismét a szabályt:

```
> Limit(D(f)(x)/D(g)(x),x=infinity)=Limit((D@@2)(f)(x)/(D@@2)(g)
  (x),x=infinity);
```

$$\lim_{x \to \infty} \left(\frac{1}{2}\,\frac{e^x}{x}\right) = \lim_{x \to \infty} \left(\frac{1}{2}\,e^x\right) \tag{2.45}$$

Most már egyértelmûen látszik, hogy a határérték ∞:

```
> Limit(f(x)/g(x),x=infinity)=limit((D@@2)(f)(x)/(D@@2)(g)(x),x=
  infinity);
```

$$\lim_{x \to \infty} \left(\frac{e^x}{x^2}\right) = \infty \tag{2.46}$$

Egyéb típusok

A $0\,\infty$, $\infty - \infty$, 1^{∞}, 0^0, ∞^0 típusú határértékek alkalmas átalakítással visszavezethetõk $\frac{0}{0}$ vagy $\frac{\infty}{\infty}$ alakú határérétkekre. Ezekre nézzünk most néhány példát:
1. $0\,\infty$ típus.
Példa
Számítsuk ki a

$$\lim_{x \to 0+} (x \ln(x))$$

határértéket!
Megoldás
Az elsõ tényezõ 0-hoz, a második $-\infty$-be tart. Alakítsuk át a kifejezést:

$$x \ln(x) = \frac{\ln(x)}{\frac{1}{x}},$$

a kapott kifejezés már $\frac{\infty}{\infty}$ alakú, és mind a számláló, mind a nevezõ teljesíti a L'Hospital szabály feltételeit. Alkalmazzuk a szabályt:

```
> Limit(x*ln(x),x=0,right)=Limit('ln(x)'/''(1/x)'',x=0,right);
```

(3.1)

$$\lim_{x\to 0+} (x\ln(x)) = \lim_{x\to 0+} \left(\frac{\ln(x)}{'\frac{1}{x}'}\right) \tag{3.1}$$

```
> Limit('ln(x)'/''(1/x)'',x=0,right)=limit(diff(ln(x),x)/diff(1/x,
x),x=0,right);
```

$$\lim_{x\to 0+} \left(\frac{\ln(x)}{'\frac{1}{x}'}\right) = 0 \tag{3.2}$$

```
> Limit(x*ln(x),x=0,right)=limit(diff(ln(x),x)/diff(1/x,x),x=0,
right);
```

$$\lim_{x\to 0+} (x\ln(x)) = 0 \tag{3.3}$$

2. $\infty - \infty$ típus.

Példa

Számítsuk ki a

$$\lim_{x\to 0} \left(\frac{1}{\sin(x)} - \frac{1}{x}\right)$$

határértéket!

Megoldás

A kifejezés valóban $\infty - \infty$ típusú, amint ezt a kisebbítendõ és a kivonandó együttes ábrája is mutatja:

```
> plot([1/sin(x),1/x],x=-2..2,-10..10,color=[blue,red],thickness=
2);
```



Alakítsuk át a függvényt, hozzuk a törteket közös nevezőre:

```
> kifejezes=normal(1/sin(x)-1/x);
```

$$kifejezes = \frac{x - \sin(x)}{\sin(x)\, x} \tag{3.4}$$

A kapott kifejezés már láthatóan $\frac{0}{0}$ alakú, és teljesíti a tétel feltételeit. Számítsuk ki a határértéket!

```
> Limit(1/sin(x)-1/x,x=0)=Limit((x-sin(x))/(sin(x)*x),x=0);
```

$$\lim_{x \to 0} \left(\frac{1}{\sin(x)} - \frac{1}{x} \right) = \lim_{x \to 0} \left(\frac{x - \sin(x)}{\sin(x)\, x} \right) \tag{3.5}$$

```
> Limit((x-sin(x))/(sin(x)*x),x = 0)=Limit(diff(x-sin(x),x)/diff
  (sin(x)*x,x),x=0);
```

$$\lim_{x \to 0} \left(\frac{x - \sin(x)}{\sin(x)\, x} \right) = \lim_{x \to 0} \left(\frac{1 - \cos(x)}{\cos(x)\, x + \sin(x)} \right) \tag{3.6}$$

A kapott kifejezés is $\frac{0}{0}$ alakú, alkalmazzuk ismét a L'Hospital szabályt (Vizsgálja meg az olvasó, hogy a feltételek teljesülnek!)

```
> Limit(diff(x-sin(x),x)/diff(sin(x)*x,x),x=0)=Limit(diff(x-sin(x),x$2)/diff(sin(x)*x,x$2),x=0);
```

$$\lim_{x\to 0}\left(\frac{1-\cos(x)}{\cos(x)\,x+\sin(x)}\right)=\lim_{x\to 0}\left(\frac{\sin(x)}{-\sin(x)\,x+2\cos(x)}\right) \tag{3.7}$$

A jobboldal számlálója 0-hoz, nevezõje 1-hez tart, midõn az x 0-hoz tart, így a keresett határérték:

```
> Limit(1/sin(x)-1/x,x=0)=limit(diff(x-sin(x),x$2)/diff(sin(x)*x,x$2),x=0);
```

$$\lim_{x\to 0}\left(\frac{1}{\sin(x)}-\frac{1}{x}\right)=0 \tag{3.8}$$

```
>
```

3. A 1^{∞}, 0^0, ∞^0 típusú kifejezések a következõképpen vezethetõk vissza az alapesetre. Tegyük föl, hogy a szóbajövõ x értékekre $0<f(x)$. Ekkor

$$f(x)^{g(x)}=\left(e^{\ln(f(x))}\right)^{g(x)}=e^{(g(x)\ln(f(x)))}.$$

A kitevõ határértékét kiszámítva, az eredeti határérték is adódik.

Példa

Számítsuk ki a

$$\lim_{x\to 0+} x^{\sin(x)}$$

határértéket!

Megoldás

Alakítsuk át a függvényt az elõzõkben leírtak szerint:

$$x^{\sin(x)}=e^{(\sin(x)\ln(x))},$$

így

$$\lim_{x\to 0+} x^{\sin(x)}=\lim_{x\to 0+} e^{(\sin(x)\ln(x))}.$$

Számítsuk ki a kitevõ határértékét:

$$\lim_{x\to 0+}(\sin(x)\ln(x))=\lim_{x\to 0+}\left(\frac{\ln(x)}{\frac{1}{\sin(x)}}\right)=$$

$$\lim_{x\to 0+}\left(\frac{1}{-x\,\frac{\cos(x)}{\sin(x)^2}}\right)=\lim_{x\to 0+}\left(-\frac{\sin(x)^2}{x\cos(x)}\right)=$$

$$\lim_{x\to 0+}\left(-\frac{2\sin(x)\cos(x)}{\cos(x)-x\sin(x)}\right)=\frac{0}{1}=0.$$

Ezért:

$$\lim_{x\to 0+} x^{\sin(x)}=\lim_{x\to 0+} e^{(\sin(x)\ln(x))}=e^0=1$$

Kérdések, feladatok

Ellenõrzõ kérdések

1. Milyen alakú határértékek kiszámítására alkalmazható -közvetlenül - a L'Hospital-szabály?
2. Milyen feltételei vannak a szabály alkalmazásának?
3. Milyen alakú határértékek számíthatók ki átalakítás után?

4. Hogyan vezethetõk vissza az 1^{∞}, 0^0, ∞^0 típusú kifejezések az alapesetre?

5. Mi a teendõ, ha ismét $\frac{0}{0}$ vagy $\frac{\infty}{\infty}$ alakra jutunk?

Gyakorló feladatok

Számítsuk ki a következõ határértékeket a L'Hospital-szabállyal! Minden esetben ellenõrizzük a feltételek teljesülését!

1. $\lim_{x\to 0}\left(\frac{e^x-1}{\sin(x)}\right)$; 2. $\lim_{x\to 0}\left(\frac{\ln(\cos(x))}{x}\right)$; 3. $\lim_{x\to 0}\left(\frac{x-\arctan(x)}{x^3}\right)$; 4. $\lim_{x\to 0}\left(\frac{e^{\tan(x)}-e^x}{\tan(x)-x}\right)$;

5. $\lim_{x\to\infty}\left(x^2 e^{(-x)}\right)$; 6. $\lim_{x\to 1}\left(\frac{x}{x-1}-\frac{1}{\ln(x)}\right)$; 7. $\lim_{x\to\infty}\left(\frac{\ln(x)}{x^n}\right)$ (n természetes szám); 8. $\lim_{x\to 1} x^{\left(\frac{1}{x-1}\right)}$.